
PDP-8の再評価と再構成

和田 英一*1

概要プログラミングの楽しみ方のひとつは, アセンブリ言語を使って昔の計算機の機械語プロ
グラムを書くことである. そういう対象になる計算機は少なからず存在するが, 代表として
とりあげて異存がないと思われるのは, 1960 年代後半から 5 万台超を出荷した DEC(Digital

Equipment Corporation)社の PDP-8( Programmed Data Processor)であろう.

PDP-8 の愛好者はいまだに多く, ウェブ上に様々な情報が散見される. ソフトウェアによるシ
ミュレータもあり, また PDP-8の通りに動くハードウェアを作ったという報告もある.

私もかなり前から PDP-8のシミュレータを作り, 簡単なプログラムを走らせたりしてきたが, 最
近またいろいろなプログラムを書いてみた結果, 改めて PDP-8のアーキテクチャの功罪を評価
できるほどになった.

一方, 単なるシミュレータ上だけでなく, 更に実感のある PDP-8 を手元に置きたいと思うよう
になり, ネット上に Raspberry Pi を使った PDP-8 の実装経験を見たこともあって, 自分でも
Raspberry Piにいくらかの入出力装置を付けた個人用 PDP-8を構成しようと思いたった.

これまで Schemeで書いてあった PDP-8のシミュレータやアセンブラは Raspberry Piで動く
Pythonに書きなおした.

Raspberry Pi の GPIO(General-Purpose Input/Output) の入出力ピンは 26 本. PDP-8 のパ
ネルのライト (12x2) とスイッチ (12+7) の個数を処理するには足りないので, Port Expander

MCP23017を利用する. MCP23017を使うと GPIOピンは 2本減るが, ピンはMCP23017の
1個あたり 16本増えるので, 2個使えば十分である.

TX-0*2

私が初めて MITを訪れたのは 1958年秋であった. い
ろいろなところを見学したが, 強烈に覚えているのは
TX-0の計算機室である.

TX-0 は MIT の Lincoln Laboratory で試作したトラ
ンジスタ計算機で, サイクルタイム 6μ秒. 1語 18ビッ
トで 4096 語の記憶装置を持つ. 命令語も 18 ビット
だが, 命令部分が 2 ビット, 番地部分が 16 ビットで
あった. 命令部が 2 ビットなので, 命令は 4 種類しか
なく, 00は sto x, 01は add x, 10は trn xでこれは
アキュムレータが負なら xへジャンプする. 最後の 11

は opr x で, 後続のビット構成でアキュムレータや入
出力機器を操作する.

後年この TX-0を設計仕直し商用化したのが PDP-1といわれている.

TX-0がその後どうなったかは知らぬが, 私が 1973年から 74年にMITの電気工学科に滞在した折, きれいな
roff出力のレポートを提出した学生が TX-0を使ったといったから, 学生が使える計算機になっていたのであ
ろう.

*1 IIJ技術研究所
ew@iij.ad.jp

*2 https://en.wikipedia.org/wiki/TX-0

PDP-8

MIT からスピンアウトして設立されたひとつが Digital Equipment Corporation (DEC) で, PDP-1 の後も
後継機を販売していた. その中でミニコンとして成功したのが PDP-8である [0].

私は PDP-8 を直接使ったことはないが, 私のいた東大計数工学科には DEC の Link-12 があり, これは
LINC(Laboratory INstrument Computer)と PDP-8の合いの子であったから, 私は専ら PDP-8モードにし
て使い, ユニークなアーキテクチャの機械を楽しんだ.

Wikipediaを見ると, PDP-8の初出荷は 1965年 3月で, 各種のモデルを合わせ 5万台以上が売れ, 廉価版の
PDP-8/Sは 1万ドル以下だったそうだ.

今でも PDP-8の愛好者は少くなく, ネット上に様々な情報がある (後述). なにしろ極めてすっきり美しいアー
キテクチャなので, シミュレータを作るのは容易で, 私も何度作ったか分らない.

PDP-8のアーキテクチャ*3

簡単に概要を示すと, 1 語は 12 ビット, 記憶装置は 12

ビットの 4096 語. 1 ページが 128 語の 32 ページで構
成される.

演算装置の中心は 12 ビットのアキュムレータで, 2 の
補数 (4096の補数)で演算される. アキュムレータの左
に 1ビットのリンクがあり, 加算の結果繰上がりがある
と, リンクのビットは反転される. アキュムレータのシ
フトは回転シフトであり, リンクを経由する. アキュム
レータのクリアはリンクに影響しない.

図の 2段目は記憶装置参照命令で 12ビット. 下のよう
な構成である.

左端の 3 ビットが命令コード. 右端の 7 ビットがペー
ジ内アドレスである. ビット 4 が 0 の時は第 0 ページ
の番地を示し, 1の時はその命令のあるページ内でのア
ドレスを示す.

ビット 3が 1の時は, その語の内容を番地とする (間接番地.)

記憶場所の八進法で 0010から 0017までを間接番地で参照すると, この番地の内容を 1増やしてから使われる
から要注意である.

記憶場所参照命令は, 命令コードの 0から 5が次のようである.

0 AND

1 TAD

2 ISZ

3 DCA

4 JMS

5 JMP

ANDは番地の指す記憶場所の内容とアキュムレータのビット毎の ANDをとり, 結果をアキュムレータに置く.

TADは番地の指す記憶場所の内容をアキュムレータに加える. 繰上がりはリンクを反転する.

*3 https://sites.fas.harvard.edu/~cscie287/fall2017/slides/

PDP-8%20Introduction%20to%20Programming%20pp.%202-4%20to%202-27.pdf

ISZは番地の指す記憶場所の内容を 1増やし, その結果が 0になれば, 次の命令をスキップする.

DCAはアキュムレータの内容を番地の指す記憶場所に入れ, アキュムレータをクリアする.

JMSはこの命令の次の番地を, 番地の指す記憶場所に入れ, その次の命令に進む. サブルーチンジャンプ.

JMPは番地の指す記憶場所の命令に進む.

命令コード 6は入出力命令で, テレタイプのキーボードから読んだり, プリンタに印字したり, 紙テープリーダ
から読み込んだりする. この命令の説明は省略する.

命令コード 7はマイクロプログラム命令で, 図に示すようにビット 3の 0, 1に従って, グループ 1とグループ
2とがある.

グループ 1はビットの 4,5,6,7が 1だと, アキュムレータをクリア (CLA), リンクをクリア (CLL), アキュムレー
タの各ビットを反転 (CMA), リンクを反転 (CML) する.

ビット 8, 9はアキュムレータとリンクを繋げた左 (RAL)または右 (RAR)回転で, ビット 10の 0/1に従って 1

ビットか 2ビット (RTL, RTR)になる. ビット 11が 1だと, アキュムレータを 1増やす (IAC).

マイクロプログラムでは実行順が重要である. 最初がクリア, 次が反転, 次が IAC, 最後が回転となる.

グループ 2は, ビット 4が 1の時, アキュムレータのクリア (CLA), ビット 5,8が 1,0の時, アキュムレータの負
(ビット 0が 1)でスキップ (SMA), 1,1の時, アキュムレータの正でスキップ (SPA). ビット 6,8が 1,0の時, ア
キュムレータの=0でスキップ (SZA), 1,1の時, アキュムレータの �=0でスキップ (SNA). ビット 7,8が 1,0の
時, リンクの �=0でスキップ (SNL), 1,1の時, リンクの=0でスキップ (SNL)である.

ビット 9が 1の時, コンソールの 12ビットのスイッチをアキュムレータとビットごと OR する. ビット 10が 1

の時, 計算機を停止する (HLT). 実行順はまずスキップ, クリア, スイッチ読込みと停止である.

プログラムでは, CMA IACのように書くだけである. こう書くとアキュムレータがビット反転され, 1足される
ので, アキュムレータの値を負にする. PDP-8には減算命令はなく, A−Bは Bをアキュムレータに置き, CMA

IACをやってから Aを足す.

ついでだが, PDP-8のビット演算は ANDしかない. A XOR Bは A + Bから A AND Bの 2倍を引く. A OR B

は CMA(CMA(A) AND CMA(B))とする.

ところで PDP-8ではプログラムは原則八進法で書く. 格納開始番地は星印 (*)を使い
*200

のように書く. 記号アドレスの定義は
start,

のようにコンマ (,)を使う. 命令のアドレス部には記号アドレスや, この命令の番地を示すピリオド (.)や, プ
ラス, マイナス (+, -)に数を続ける.

斜線 (/) は行末までのコメントの開始を示す. プログラムの最後には ($)を書く.

例えばアキュムレータにある正を数を十進法の 10で割り, 商を q, 剰余を rに置くサブルーチン div10を八進
の 100番地からに入れるには (以後プログラムは小文字で記す.)

*100 (↗)

div10,0 /サブルーチンの入口. 帰り番地が入る
dca r /被除数を rに入れる
dca q /商の場所 qをクリアする
tad r /部分剰余をアキュムレータに置く
tad m10 /10を引く
spa /剰余が正ならスキップ
jmp .+3 /負になったので後始末へ
isz q /剰余を 1増やす (↗)

jmp .-4 /4命令前へ戻る
tad p10 /引き過ぎた剰余を戻す
dca r /剰余を rに入れる
jmp i div10 /主プログラムへ戻る

m10,-12 /定数 -10

p10,12 /定数 10

r,0 /剰余の場所
q,0 /商の場所
$

のように書く.

PDP-8の算術ルーチン

PDP-8は RISCの走りみたいなものなので, 乗除算からサブルーチンの世話になる. それらを纏めたライブラ
リがネットにある. 例えば乗算はこんな具合である.

このプログラムはなかなか超絶技巧である. 呼び出すとき乗数をアキュムレータに置き, jms multで飛ぶ. 被
乗数はこのサブルーチンジャンプ命令の次に置く. サブルーチンではまずリンクを零にする. 乗数が正ならス
キップ; 負なら cmaでアキュムレータの 1と 0を交換し, iacで 1を加え正にする. 同時にリンクを反転する.

正の乗数を mp1に置く. アキュムレータが零になったので, 積の場所 mp5を零にする.

000 mult, 0 (↗)

001 cll

002 spa /test for negative multiplier

003 cma cml iac

004 dca mp1

005 dca mp5

006 tad i mult

007 sna /test for zero mutiplier

008 jmp mpsn+2 /jmp if multiplicand=0

009 spa /test for negative multiplicand

010 cma cml iac

011 dca mp2

012 tad thir

013 dca mp3

014 mp4, tad mp1 /multiply loop proper

015 rar

016 dca mp1

017 tad mp5

018 szl /test if multiplicand should be added

019 tad mp2

020 cll rar

021 dca mp5 (↗)

022 isz mp3 /test for end of loop

023 jmp mp4

024 tad mp1

025 rar

026 mpsn, szl

027 jmp comp

028 dca mp1

029 tad mp5

030 mpz, isz mult /exit to calling program

031 jmp i mult

032 comp, cma cll iac /complement prduct

033 dca mp1

034 tad mp5

035 cma

036 szl

037 iac

038 jmp mpz

039 thir, 7764 /eleven in decimal

040 mp1, 0

041 mp5, 0

042 mp2, 0

043 mp3, 0

6行目で被乗数を読み込む. それが零なら 8行目から飛ぶ. それが負なら 10行目で正にし, リンクを反転する.

つまりリンク =1は積の負を示す (以下でこのビットを sという. 正の被乗数を mp2に置く.

カウンタ mp3に 77648 つまり −1210 を置き, 14行目からループが始まる.

14行目 (カウンタ=−12, sはリンク) 乗数を取り右回転シフト. sはビット 0に, リンクには乗数の最下位が
入る. これを mp1に置く. つまり乗数は 1桁減り, 代りに sが入った.

17行目 積 (mp5)を取り, リンクが 1なら被乗数を足す. そして右回転シフト. 積の最下位がリンクに入り, 1

ビット右にずれた積が戻る.

再び 14行目 (カウンタ=−11) 乗数を右回転シフト. sはビット 1に, 積の最下位がビット 0に. 乗数の下から
2桁目がリンクに.

こうしてカウンタ=−1で積の右回転シフトが済んだ時, sはビット 11に, 積の最下位がビット 10にある. こ
こでループを脱出し, 25行目で積を右回転する. sがリンクに入り, 積の下半が mp1に収まる. 27行目でリン
クが 1なら積の符号反転のため, compへ進む. その必要がないなら, 積の下を mp1に置き, 上をアキュムレー
タに置き, 帰り番地を 1増やして戻る.

32行目から倍長語の符号反転で, 下の語でリンクを零にし, ビットを反転し, 1を足す. この時, 語が零なら繰

上がりが出てリンクを 1にする. 次に上の語のビットを反転し, リンクが 1なら 1を足して終わる. 中々よく
出来ている.

ただ, 4000×4000は 60000000にはならず, 4000が正だと思って 20000000になってしまう.

一方, 除算は倍精度の被除数を単精度の除数で割るもので, 剰余の符号は被除数のそれに合わせる. 呼出しは,

サブルーチンジャンプ命令の次に被除数の下部分と除数を置き, 被除数の上をアキュムレータに置く.

このプログラムの中心は 33行目から 45行目のループである. 被除数の 2語を左へシフトしながら除数を引け
れば引くを繰り返す. 41行目から入った時, 28行目でスキップしているからリンクは 0だ. そこで下の被除数
を左回転するから, 最上位のビットがリンクへ行く. 33行目から今度は上の被除数を左回転し, 下からのビッ
トを繋げる. その結果をとりあえず元の場所に置く.

除数を引く (というか −除数を足す). 引ければキャリーがリンクに入る. リンクが 1なら引けたので差をしま
う. 41行目で下を取り出し, 左回転するから, 商が右から入り, 上のビットがリンクへ行く.

000 divide, 0 (↗)

001 cll

002 spa /dividend<0?

003 cma cml /yes complement and set c(l)

004 dca hdivnd /high order dividend

005 snl

006 cma

007 dca sdvnd /set dividend sign switch

008 tad i divide /fetch low order dividened

009 szl

010 cma cll iac /yes: complement

011 dca ldivnd /low order dividend

012 szl /carry?

013 isz hdivnd /yes

014 isz divide

015 tad i divide /fetch divisor

016 cll

017 sma

018 cma cml iac /negate it

019 dca divsor /save divisor

020 snl /was it <0?

021 cma /yes: ac=-1

022 tad sdvnd

023 dca snswer /answer sign switch

024 cll

025 tad divsor /compare divisor

026 tad hdivnd /with dividend

027 isz divide

028 szl cla /over flow?

029 jmp i divide /yes: divisor<dividend

030 tad m13 /13 shifts(↗)

031 dca divcnt

032 jmp dv2

033 dv3, tad hdivnd

034 ral

035 dca hdivnd /dividend left shift

036 tad hdivnd

037 tad divsor /compare divisor:dividend

038 szl

039 dca hdivnd /remainder after subtract

040 cla

041 dv2, tad ldivnd /quotient bits

042 ral /enter here

043 dca ldivnd

044 isz divcnt /done 12?

045 jmp dv3 /no: continue

046 tad hdivnd /remainder

047 isz sdvnd /dividend<0?

048 cma iac /yes

049 dca hdivnd

050 tad ldivnd /quotient

051 isz snswer /answer<0?

052 cma iac /yes: negate

053 cll

054 jmp i divide /exit

055 hdivnd, 0

056 ldivnd, 0

057 divsor, 0

058 sdvnd, 0

059 snswer, 0

060 divcnt, 0

061 m13, -15

除算で面倒なのは被除数が大き過ぎて商がオーバーフローすることのチェックである. 大雑把にいえば上の被
除数の絶対値が除数の絶対値より大きければ駄目だがそう簡単ではない. それを念頭に置き, プログラムを見
なおそう.

リンクを 0にし, 上の被除数 < 0なら 1の補数をとり, リンクを 1にする. 被除数の正負は上で決るので上か
ら始めるが, 符号反転に下からキャリーが来るのは一旦我慢する. 正の上の被除数を hdivndにしまう. 7行目

では被除数の正/負に従って, −1/0を sdvndに置く. 剰余の符号を被除数に合わせるために記憶する. 8行目
で下の被除数を取り, 必要に応じて反転し, それでキャリーが出れば上の被除数に 1を足す. 15行目で除数を
得, 負にして 19行目で divsorにしまう.

18行目で除数が正ならリンクを 1にし, そうだと 21行目に来て, アキュムレータを −1にする. それに sdvnd

を足すから, 被除数と除数のいずれか正の時, −1が snswerに入る. つまり商の正負を記憶する. (ともに正の
時は −2.)

24行目からは被除数と除数の大小比較で, 被除数が大きければ, リンクを 1にして帰る.

その後除算のループを実行し, 修了後, 先程のフラグに従って符号を定めて戻る.

このプログラムでは, iszを巧みに使っているのが分る.

書き直した乗除算ルーチン

私は新しいアーキテクチャや新しいプログラミング言語に出会ったとき, それを体験するのに乗除算と UNIX

の calのようなカレンダーのプログラムを書いてみることにしている. PDP-8の評価にもそれをやってみた.

次にそれをざっと説明する. 前述の乗算プログラムと違い, 2の補数のまま計算するのが好きなので, 今回もそ
うした. また PC-1の w命令のように, 初めに単精度の部分積 cがあって, 単精度の乗数 bと被乗数 aに対し,

倍精度の a*b+c を得るものである.

要するに, 部分積 cに乗数 bを下の桁から 1ビットずつ見ながら, 1なら被乗数を足し, 0なら足さぬで, 部分
積を右シフトするものである. 乗数の最左のビットは −1なので, その桁が 1なら被乗数を引く.

乗算の進むに従い, 乗数は短くなり, 部分積は長くなるので, 部分積の下の方は乗数の場所を使う. 倍精度の部
分積の右シフトにはリンクビットが使える. 面倒なのは右シフトのとき, 部分積の符号を正しく入れることで
ある.

左の表はアキュムレータが 3 ビットで, その左にリン
クビットがあったとして, 3ビットの 2数と足したとき
(左の表)と引いたとき (右の表)の結果 (各ますの上段)

と, なって欲しい値 (下段)である.

減算の方から見ると, 上の欄外に 4,5,6,7,0,1,2,3とある
のはビットそのものである. その下の-4,-3,-2,-1は 2の
補数のときの解釈. 左の欄外にも同様の表記がある.

右下の 0は左欄外の 3から上欄外の 3を引いた結果.

その左隣の 1は 3から 2を引いた結果で, ますの下半分が空白なのは, このまま結果を示す.

この約束で第 1 象限をみると, 負の数から正の数を引いた結果になっている. 最上段の中央の右の 4 は, 4 の
ビット列から 0 のビット列を引き, 4 のビット列になったが (上段), 算術演算の解釈としては, -4 から 0 を引
き, -4(L4)になって欲しいことを示す.

解釈の仕方が分って加算の表を見ると, 第 2象限と第 4象限, つまり負同士, 正同士の加算はビット列の結果の
ままでよいが, 正と負の加算では, ビット列に Lがある (リンクが 1)のときは, リンクを反転したのが演算結果
であることが分る.

すなわち, 2の補数の 2つの数を足したとき, 両方が同符号なら, リンクビットはそのままでよいが, 異符号な
ら, リンクを反転するようプログラムを書かなければならない. a+bは

cla cll /アキュムレータとリンクをリセット
tad a /aを足す
spa cla /aが正ならスキップ アキュムレータを零に
cml /リンクを反転

tad b /bを足す
spa /bが正ならスキップ
cms /リンクを反転
tad a /aを足す
rar /右回転シフト

のようにする.

次にあるのが, 私の乗算サブルーチンで, 3 行目まででカウンタを十進で −11 に設定し, ループに入る. muls

が 1桁乗算の下請けで, 13行目でアキュムレータに 1を置き, 14行目の andで乗数の最下位のビットを取り
出す. 零でなければ被乗数 r0を置く. その後が上述のリンクの扱いである. mulsの下請けが 2箇所から呼ば
れているのは, 11回の加算のあと, 1回の減算をするからである. 7行目でループを抜けたあと, 9行目までで
被乗数を負にし, もう一度下請けにいき, サブルーチンから戻る.

積の最大値は 3777*3777+3777=17774000, 最小値は −4000*3777−4000=−2000000 である.

000 mul,0 (↗)

001 cla cll

002 tad m13

003 dca r3 /set counter

004 loop, jms muls

005 isz r3 /test counter

006 jmp loop

007 tad r0 /negate muliplicand

008 cma iac

009 dca r0

010 jms muls

011 jmp i mul /return

012 muls,0 /mul subroutine

013 cla cll iac

014 and r1 /test bit

015 sza cla

016 tad r0 /test sign of multiplicand

017 spa(↗)

018 cml

019 dca r4

020 tad r2 /add partial product

021 spa

022 cml /set link

023 tad r4 /add multiplicand

024 rar /shirt right partial product

025 dca r2

026 tad r1 /shift right multiplier

027 rar

028 dca r1

029 jmp i muls

030 r0,0 /multiplicand

031 r1,0 /multiplier

032 r2,0 /partial product

033 r3,0 /counter

034 r4,0 /workspace

035 m13,-13 /-11 in decimal

除算も 2の補数のまま計算する. 2の補数の除算では, 剰余の符号は通常のアーキテクチャやプログラミングと
違い, 除数の符号に合わせるのが合理的という私の持論に従ったサブルーチンにしてある.

倍精度の被除数 n と, 零でない単精度の除数 d に対して, 単精度の商 q と単精度の剰余 r を得る. 除算だか
ら当然オーバーフローがあり得る. 単精度が 12 ビットの場合, 正の除数は 1 ≤ d ≤ 2n−1 − 1, 負の除数は
−2n−1 ≤ −1の範囲にあり, 除数が正のときの剰余は 0 ≤ r < d, 負のときの剰余は d ≤ r < 0である.

除算では先ず除数を負にしたものを用意する. 引き放し除算では, 部分剰余と除数が同符号なら除数を引き, 異
符号なら足す. 減算命令のない PDP-8では, 減算に備えて負の除数が必要になるので, それを作るのである.

6行目から 12行目で倍精度の被除数を左 1ビットシフトし, 19行までで除数を引くか足すする. この演算の結
果, 被除数の符号が変らなければ, この除算はオーバーフローする. オーバーフローのテストは 12行目辺りに
入れる必要があるが, 今はさぼっている.

23行目でカウンタを設定したあと, r0と r1を左シフトしながら, 部分剰余の符号を見ながら除数の加減算を
続ける. 引くときには r0の右端から商の 1のビットを挿入する.

テストデータで実行した結果は次のようだ. 私の Happy Hacking Calculator*4 の octal modeで検算すると,

*4 http://www.iijlab.net: ew/hhcmanualen.pdf

被除数 除数 商 剰余
a 1777 3777 3777 3777 3776

b 1777 7777 4000 4000 7777

c 6000 4000 3777 4000 0000

d 6000 0000 4000 3777 4000

a は 1777 3777 ent 3777 / 3777 pop 3776. b は 1777 7777 ent 4000 chs

−4000 / −4000 chs 4000 pop −1 10000 + 7777. c は 6000 4000 ent

100000000 − −17774000 3777 / −4000 10000 + 4000 pop 0. d は 3777

4000 ent 100000000 − −20000000 4000 ent 10000 − −4000 / 3777 pop

−4000 10000 + 4000.

となって一応合っていることが分る. ビットパターンで出力するサブルーチンを使って表示しているから変に
見えるが安心してよい.

000 div,0 (↗)

001 cla cll

002 tad r2

003 cma iac

004 dca r3 /neg divisor

005 cll

006 tad r0

007 tad r0

008 dca r0 /r0 left shift

009 ral

010 tad r1

011 tad r1 /r1 left shift

012 dca r1 /need overflow check

013 rar /link to acc

014 tad r2

015 spa cla /skip if same sign

016 jmp .+3

017 tad r3 /same sign sub

018 jmp .+2

019 tad r2 /diff sign add

020 tad r1

021 dca r1

022 tad m13 /init counter

023 dca r4

024 loop, cll cla

025 tad r0 /left shift r0

026 tad r0

027 dca r0

028 ral

029 tad r1 /left shift r1

030 tad r1

031 dca r1

032 rar(↗)

033 tad r2

034 spa cla /skip same sign

035 jmp .+6

036 iac

037 tad r0 /same set quotient bit

038 dca r0

039 tad r3 /sub

040 jmp .+2

041 tad r2 /diff sign add

042 tad r1

043 dca r1

044 isz r4

045 jmp loop

046 cll cla cml rar /4000 ac

047 and r1

048 tad r2 /compare signs

049 spa cla

050 jmp .+3 /jump diff sign

051 iac /prepare 1

052 jmp .+4

053 tad r1 /diff sign

054 tad r2

055 dca r1

056 tad r0 /r0*2+0/1

057 tad r0

058 dca r0

059 jmp i div

060 r0,0 /lower dividend, quotient

061 r1,0 /higher dividend, remainder

062 r2,0 /divisor

063 r3,0 /negative divisor

064 r4,0 /counter

065 m13,-13

calのプログラムも完成しているが, ここでは割愛する. PDP-8の少ない命令とレジスタを使ってアルゴリズ
ムを考案するのは, それ自体が楽しいが, 現時点ではノスタルジーでしかないのも十分理解している.

PDP-8好きはあちこちに

左上はMac用の PDP/8Eシミュレータである. テレタイプの音まで聞こえるという.

右上はシングルチップの PDP-8. 左下は私が昔作った PDP-8 のシミュレータで, PDP-8 の 1 語 12 ビット,

4096語の記憶装置を示している. 収まっているのは, van der Poel 先生の Lispシミュレータで, 右半分のヒー
プ領域が見える [1]. 右下は Raspberry Piで駆動する PDP-8Iで, 実機の 2/3のサイズらしい [2].

自分の PDP-8を作る
PDP-8のシミュレータを私は C, Scheme, Pythonで書いたりした.

だから PDP-8 のプログラムはすぐに走らせてみることが出来るが,

やはりライトが点滅する様子も見たい. ネットを探すと作ってみたと
いう記事が見付かる. すこし考えてみると, Raspberry Piで Python

のシミュレータを動かし, Raspberry PiのGPIOで 12個×2のライ
トを点け, 12 個 + アルファのスイッチを読み込めば, ある程度の実
感が得られそうである. GPIOのピンの数は 26本だが, MCP23017

を使えば 1個で 16本に増やせるから, 23017を 4個用意すれば出来
そうである. PDP-8の入出力は 33型テレタイプであったが, そうい
うものは既にないので, パソコンなどで代用する.

Raspberry Pi 2 Modle B

MCP23017

PDP-8Eの面構えで参考にしたのは次の図である.

これを見て起した正面パネルは次の図のようだ (次ページ.) 元の顏を見比べてほしい.

自家用 PDP-8の操作パネルに搭載するライトとスィッチは, 元の PDP-8にあるもの全部をつける必要はない
ので, とりあえず次の図にあるものにした.

LEDによるライトは 12ビットで 2段. 上段がプログラムカウンタで下段がアキュムレータである. 記憶装置
を読み出す時は, 上段が番地で下段がその内容を示す. リンクが表示できないのは不便かもしれないが, 本家の
PDP-8では下段に切替えスィッチがあって, STATUS位置にするとビット 0がリンクを表示する. その時, 他の
ビットは別の状態を示すが, その辺は無視するのでリンク表示も割愛されてしまった.

スィッチは, スィッチレジスタに対応する 12個のものの他, その右手にある 7個を用意する. ADDRはスィッチ
レジスタの設定を CPMA(Central Processor Memory Address) レジスタへ転送するものである. その右に隙
間があるが, ここには拡張記憶装置用の EXTDというスィッチがあるが, 今回はそれもないのでそのスィッチも
省略した.

CLRはレジスタ類をクリアして走行状態にするもの, CONTはクリアせずに走行を再開するものである.

走行状態の時は RUNのライトが点灯する.

EXAMは CPMAの差す記憶場所の内容を下段のライトに示し, CPMAの内容を 1増やす.

HALTは走行している計算機を停止する. RUNライトは消える.

STEPはこれを押すと, プログラムを 1ステップずつ実行する. DEPはデポジットキーで, 上段のライトの示す
CPMAの場所に, スィッチレジスタの内容を転送する. EXAMと同様で, CPMAの値は毎回 1ずつ増える. こ
れらのスイッチには作業用として番号もつけてある.

合わせるとライトが 12+12+2(リンクと run)=26. スィッチが 12+7=19. 計 45個が必要になる. Raspberry

Piの GPIOのピンは元々 26本あるが, MPC23017に 2本使うので 24本. MPC23017を 2個採用すると, 1

個当たり 16本あるから 32本. この 56本のうち 45本を使うことになる.

部品
トグルスイッチ ミヤマ
MS500 BB ON-(ON) 7個 200*7=1400 1400*1.08=1512円
MS500 KB ON-OFF 12個 170*12=2040 2040*1.08=2203円
抵抗内蔵 LED I-06249 3個 240*3=720円
LEDブラケット P-11465 3個 150*3=450円

1512+2203+1170=4885円
スイッチサイエンス MCP23017 Raspberry Pi用 GPIOエクスパンダ 972*2+150=2094円
Raspberry Pi B+用 T型 I/O延長基盤 837+150=987円

最初の設計の全体図はこのようだ. アルミのパネルは木製の台に取り付ける. その木製の台にはブレッドボー
ドを 2,3枚取り付け, その上で配線する.

上の図は最初の案で, 中央のブレッドボードに T型延長基盤を載せ, 左右のブレッドボードに GPIOエキスパ
ンダを載せる. GPIOとエキスパンダ間の配線は次のようだ. GPIOのピンの配置とMCP23017のピンの配

置を次に示す.

ライトは抵抗内蔵の LEDをMCP23017のピンと GNDの間
に接続し, スイッチは GPIOの端子と 3.3vの間に pull down

接続したので, 抵抗類の配線が不要になった.

最初の案では 3 枚であったブ
レッドボードは, その後の検
討の結果, 2 枚で足りることが
分り, ブレッドボードからはみ
出していた T 型延長基盤もブ
レッドボード内に収まること
になった.

左の図がそのブレッドボード
の配線である. 実際にはこのよ
うに横に並ぶのではなく, 右の
ボードが左のボードの上につ
ながる. (最後のページの写真
を参照)

これらのブレッドボードに差さる延長基盤とエクスパンダは次の図のようなものである.

Raspberry Pi B+用 T型 I/O延長基板

MCP23017 Raspberry Pi用
GPIOエクスパンダ

I2C

Inter-Integrated Circuit

GPIOとMCP23017との間の通信には, I2Cというプロトコルを使う. 実はこれを使うので, PDP-8のシミュ
レーションは予想外に遅くなることが分った.

I2Cでは SDA (serial data)と SCL (serial clock)の 2本を使う. これは GPIOの 2番と 3番のピンを使うこ
とになっている. 通信のビットレートは 100KHzと 400KHz. もっと高速なのもあるが, このシミュレータで
は 100KHzのまま使っている.

2 個あるエクスパンダには 0x20 と 0x21 のアドレスを付ける. それぞれのエクスパンダで A0～A7, B0～B7

の 16個の I/Oが制御出来る. それぞれのエクスパンダには 0x00, 0x01, 0x12, 0x13, 0x14, 0x15 のアドレス
を持つレジスタがあり, GPIOはそれらのレジスタと情報をやりとりする.

bus.write byte data(dev, outreg, data)で devの outregへ dataを書き,

data=bus.read byte data(dev, inreg)で devの inregから dataを読む.

今回の入力は GPIOから直接なので, この機能は使わない.

LEDの制御

Rapberry Piで動く PDP-8のシミュレータは, 前述のように Pythonで記述してある. 命令のフェッチ, 実行
のループから状態を LED で表示するルーチンを呼ぶ. それがこの LED 制御のプログラムである. ブレッド

ボードの図にあるように, 左端にあるアキュムレータの 0のビットが, エキスパンダのレジスタの 0で右側に
あるからまずビットの左右反転をする. ここは設計ミスというべきである.

下のプログラムの先頭の bitrevが 12ビット反転で, Pythonだとこんな風になる. その下の displayがシミュ
レータから呼ばれるプログラムで, グローバル変数, lacc(リンクとアキュムレータ, 13 ビット), pc(プログラ
ムカウンタ, 12ビット), run(1ビット)を受け取り, マスクしたりシフトしたりして, レジスタ用のパターンを
作っている.

def bitrev(n): #12bit reverse

return int(’{:012b}’.format(n)[::-1],2)

def display():

global lacc,pc,run

rev=bitrev(pc) #display pc and run

areg=(rev&0o77)<<2

breg=(rev&0o7700)>>6

bus.write_byte_data(0x20,0x14,areg)

bus.write_byte_data(0x20,0x15,breg|(run<<6))

rev=bitrev(lacc&0xfff) #display acc and link

areg=(rev&0o77)<<2

breg=(rev&0o7700)>>6

bus.write_byte_data(0x21,0x14,areg|(lacc>>12)<<1)

bus.write_byte_data(0x21,0x15,breg)

rev(PC) BA9876543210

breg 7654321076543210 areg

R

rev(AC) BA9876543210

breg 7654321076543210 areg

L

テストプログラム

シミュレータが出来たので, 走らせるプログラムを用意しなければならない. とりあえず, Pythonでインタプ
リタを書く必要があるが, PDP-8のアーキテクチャが簡単なので, それは難なく完成した.

PDP-8で走るプログラムには, 前述の Lispがあるが, それ以外にも欲しい. 今回用意したのは, 素因数分解の
プログラムで, 23ビット (倍精度で正の数)を素因数分解するものである.

それには倍精度被除数の除算の剰余を得るサブルーチンと, 倍精度の正数の平方根をとるサブルーチンが必要
である. 下に示すプログラムで, (rem n d)と (sqrt n)は除算を使わない剰余と開平を Schemeで試みたも
のである. これで上手くいくことが分ったので, PDP-8のアセンブリ言語でプログラムした.

(define (rem n d) ;nを dで割った剰余を計算する
(let ((s 0))

(set! d (do ((e d (* e 2))) ((>= e n) e) ;dを d>=nになるまで 2倍し
(display (list e)) (set! s (+ s 1)))) ;その回数を sとする
(display (list d))

(do ((s s (- s 1))) ((or (< s 0) (= n 0)) n) ;sが負になるか nが 0になるまで
(display (list n d))

(if (> n 0) (set! n (- n d)) ;nが正なら n ← n−sとし
(set! n (+ n d))) ;nが負なら n ← n+sとする

(set! d (/ d 2))))) ;dを 2で割る
(rem 100 8)(8)(16)(32)(64)(128)(100 128)(-28 64)(36 32)(4 16)(-12 8)=> -4

(rem 100 4)(4)(8)(16)(32)(64)(128)(100 128)(-28 64)(36 32)(4 16)(-12 8)

(-4 4)=> 0 ;実行例
(rem 96 4)(4)(8)(16)(32)(64)(128)(96 128)(-32 64)(32 32)=> 0

(define (sqrt n) ;nの平方根を計算する
(let ((q 0)) ;平方根の初期値
(do ((s 22 (- s 2))) ((< s 0)) ;sを 22から 2ずつ減らす
(let ((a (fix:lsh (+ (* q 4) 1) s))) ;4q+1を s桁左シフトし aとする
(if (>= n a) (begin (set! q (+ q q 1)) (set! n (- n a)))

(set! q (+ q q))) ;n>=aなら q ← 2q+1, n ← n−aとし
(display (list a n q)))) ;n<aなら q ← 2qとする

q))

(sqrt 99) (1024 99 0)(256 99 0)(64 35 1)(80 35 2)(36 35 4)(17 18 9)=> 9

(sqrt 100) (1024 100 0)(256 100 0)(64 36 1)(80 36 2)(36 0 5)(21 0 10)=> 10

(define (primetest n) ;奇数 nの素数性を調べる
(let ((a (floor->exact (sqrt n))) ;nの平方根を aとする

(dd ’(1 2 2 4 2 4 2 4 6 2 6 4 2 4 6 6 2 6 4 2 6 4 6 8 4 2 4

2 4 8 6 4 6 2 4 6 2 6 6 4 2 4 6 2 6 4 2 4 2 10 2 10))

(ind 0) (d 2)) ;ddは 2,3,5,7で割れない疑似素数列の差, dは試す除数
(let loop ()

(set! d (+ d (list-ref dd ind))) ;次の除数を作る
(if (> d a) (display (list ’prime n)) ;dが aを超えたら素数
(if (= (modulo n d) 0) (display (list ’composite n));割り切れたら合成数
(begin (set! ind (+ ind 1)) ;ddの列を循環して除数を作る用意
(if (= ind 52) (set! ind 4))

(loop))))))) ;loopへ戻る

夏のプロシンの時点では, ハードウェアはすべて完成していたが, シミュレータのスイッチ読込みプログラム部
分が未完であった. また時間のあるときに, その辺のプログラムを追加し, この名機のブートの気分を味わいた
いと考えている.

討論

Q. 竹内 シミュレータの速度はどうか.

A. ソフトウェアのシミュレータは実機よりも遥かに速い. しかし I2Cのシリアル通信は相当遲い. 速くする方
法はあるが, 速度には関心がないので変更してはいない.

Q. 無名氏 TX-0は何台あったか.

A. 試作機なので 1台だと思う.

写真

右 上から見た様子
下 正面パネル
下の下 GPIOエクスパンダ
下の右 T型 I/O延長基盤の付近

